```
log using "22_analysis_log.txt", text replace
//// RCT Survey: Analysis
//// Lucas Reddinger <jlr@lucasreddinger.com>, David Levine, Gary Charness
//// 2021 April 24

cls
clear
set more off

cd "E:\lucas\projects_private_git\academia\projects\vaccine\rct"

use "data\data_shuffled_cleaned_2021-04-23_0939.dta", clear

**** Drop some variables that we're not using at this point in time
drop *_pagesubmit

**********************
**** Generate variables
**********************

**** Define segment and control variable lists
local segments "d_b d_l d_c d_r d_p"
local indivchars "d_male d_age_lt45 d_age_gt65 d_smalltown_rural"

**** Recommendation indicators
gen m_rec_b = ( m_rec_lebron | m_rec_kizzy | m_rec_obama | m_rec_warnock )
gen m_rec_l = ( m_rec_fernandez | m_rec_jlo | m_rec_badbunny )
gen m_rec_c = ( m_rec_trump )
gen m_rec_r = ( m_rec_warnock | m_rec_pope | m_rec_rickwarren )
gen m_rec_b_r = ( m_rec_b & m_rec_r ) // This is the only non-null intersection

**** Concordant recommendation
// Notice that here we subtract off m_rec_b_r if both d_b and d_r.
// That is, If a Black religious person gets a Warnock message, it obtains 1 point, not 2.
// Accordingly, concordant_rec is either 0 if not concordant or 1 if concordant.
gen concord_recommended = d_b*m_rec_b + d_l*m_rec_l + d_c*m_rec_c + d_r*m_rec_r - d_b*d_r*m_rec_b_r
tab concord_recommended

**** Concordant by message element
// Note that we need not multiply each message indicator by the demographic
// indicator because the messages were targeted to only those demographics.
// The one exception is m_frdm_c_r; this was shown to (d_c | d_r), but here we
// only count it as concordant if (d_c & d_r), so we use m_frdm_c_r_ANDdcANDdr.
gen concord_commimpact = m_comm_b + m_comm_l + m_comm_c + m_chrc_r
gen concord_poptested = m_popl_b + m_popl_l
gen concord_children = m_chld_p + m_chld_p_l
gen concord_elders = m_eldr_c + m_eldr_r + m_eldr_p
gen concord_protection = m_prtx_c
gen concord_gatherings = m_gthr_l + m_gthr_r + m_gthr_p + m_frdm_c_r_ANDdcANDdr
gen concord_availability = m_avyl_r + m_avyl_p

**** Count total concordant messages and recommendation
gen num_concord_msg = ///
      concord_commimpact + concord_poptested + concord_children ///
    + concord_elders + concord_protection + concord_gatherings ///
    + concord_availability + concord_recommended

**** Create concordant score
// Give one point for each concordant message received.
// Give an extra point if Latinx and treated with Spanish parallel text.
// (Note that only Latinx receieves Spanish, so no need to multiply by d_l.)
gen concordant_score = num_concord_msg + t_spanish

**** Concordant messages by segment
** Black segment: community impact, population tested, recommendation
gen concord_b = m_comm_b + m_popl_b + d_b*m_rec_b
** Latinx segment: comm. impact, pop. tested, gatherings, recommendation
// We do not include m_chld_p_l because it was only a modification of m_chld_p
// for Latinx; the Latinx segment did not obtain a control of m_chld_p.
gen concord_l = m_comm_l + m_popl_l + m_gthr_l + d_l*m_rec_l
** Conservative segment: comm. impact, elders, protection, freedom
// Again here we only count the m_frdm_c_r message as concordant if the subject
// is both conservative and religious.
gen concord_c = m_comm_c + m_eldr_c + m_prtx_c + m_frdm_c_r_ANDdcANDdr
** Religious segment: comm. impact (church), elders, gatherings, avail., rec.
// To avoid double-counting m_frdm_c_r, we omit it from the religious segment
// and only include it in the conservative segment above.
gen concord_r = m_chrc_r + m_eldr_r + m_gthr_r + m_avyl_r + d_r*m_rec_r
```

```
** Parent segment: children, elders, gatherings, availability
// We can simply use (m_chld_p + m_chld_p_l) because no one received both.
gen concord_p = m_chld_p + m_chld_p_l + m_eldr_p + m_gthr_p + m_avyl_p

*****************************
**** Drop useless observations
*****************************

// drop outcomes of "Don't know / Prefer not to say"
drop if vax_likely==8

************************
**** Drop subsegments with 10 or fewer obs.
************************

tab subseg

bysort subseg: gen subseg_N = _N
drop if subseg_N <= 10

tab subseg

*************
*************
**** ANALYSIS
*************
*************

****************
**** Descriptive
****************

**** Descriptive tables
// We'll produce many more tables than those here
tab d_relig d_r

**** Descriptive regressions on segment
ologit vax_likely `segments'
ologit vax_likely `segments' `indivchars'

**** Descriptive regressions on segment
ologit vax_likely i.subseg
ologit vax_likely i.subseg `indivchars'

*********************
**** Primary analysis
*********************

**** Likeliness to vaccinate = F(concordant_score)

reg vax_likely concordant_score i.subseg, cluster(subseg)
eststo e1

ologit vax_likely concordant_score i.subseg, cluster(subseg)
// Average marginal effects using margins
margins, dydx(concordant_score)
eststo e2

esttab e1 e2 using "30_results.txt", ci label addnote("Likelihood 1 to 7") append ///
    title("How does adding an additional concordant message affect likelihood to vaccinate?")
eststo clear

** Typical change in Pr(each response) as concordant_score rises
// https://www3.nd.edu/~rwilliam/xsoc73994/Margins05.pdf
*findit spost13_ado

** Shows the effect of one more concordant_score point on Pr(each value of vax_likely)
// Due to a bug, we have to remove the variable's label (we add it back afterward)
label values vax_likely .
mtable, dydx(concordant_score)
label values vax_likely vax_likely_label

**** Likeliness to vaccinate = F(concordant score by topic)
reg vax_likely concord_commimpact concord_poptested concord_children ///
            concord_elders concord_protection concord_gatherings ///
            concord_availability concord_recommended i.subseg, cluster(subseg)
est sto e1

```

```
157   **** Likeliness to vaccinate = F(concordant score by segment)
158   reg vax_likely concord_b concord_l concord_c concord_r concord_p i.subseg, cluster(subseg)
159   est sto e2
160
161   esttab e1 e2 using "30_results.txt", ci label addnote("Likelhood 1 to 7") append ///
162       title("Which concordant topics matter? Which subsegments respond to concordance?")
163   eststo clear
164
165   **************
166   **** Close log
167   **************
168
169   log close
170
```